

6 Finding a Function When Its Derivative Is Known

In Chapter 3 we motivated the definition of the definite integral by showing how the integral can be used to give the exact distance traveled when one knows the instantaneous velocity for all times. In this chapter we will study the general problem of constructing a function from its derivative.

6.1 The Definite Integral Revisited

Recall how the definite integral was defined in Chapter 3. Let f be a function which has domain $[a, b]$ and suppose that f is continuous throughout (except, possibly, at a finite number of points) and is bounded on $[a, b]$. Let $[a, b]$ be partitioned into n equally spaced subintervals $\{a = t_0 < t_1 < \cdots < t_n = b\}$. Call the width of each of the individual subintervals Δt , *i.e.*

$$\Delta t = t_1 - t_0 = t_2 - t_1 = \cdots = t_n - t_{n-1} = \frac{b - a}{n}.$$

As in Chapter 3.2 two sums are constructed: the left-hand sum and the right-hand sum,

$$leftsum = f(t_0)\Delta t + f(t_1)\Delta t + \cdots + f(t_{n-1})\Delta t = \sum_{k=0}^{n-1} f(t_k)\Delta t,$$

and

$$rightsum = f(t_1)\Delta t + f(t_2)\Delta t + \cdots + f(t_n)\Delta t = \sum_{k=1}^n f(t_k)\Delta t.$$

Examples in Chapter 3 suggest that it can be expected that, as n tends to infinity and the subdivisions become smaller, both sums may approach the same number. This common limit, if it exists, is defined to be the *definite integral*.

Definition

The **definite integral** of f from a to b , denoted by

$$\int_a^b f(t) dt,$$

and is the limit of the left-hand or right-hand sums as n tends to ∞ , *i.e.*,

$$\int_a^b f(t) dt = \lim_{n \rightarrow \infty} \left(\sum_{k=0}^{n-1} f(t_k) \Delta t \right)$$

and

$$\int_a^b f(t) dt = \lim_{n \rightarrow \infty} \left(\sum_{k=1}^n f(t_k) \Delta t \right).$$

Each of these sums is an example of what is called a *Riemann sum*, f is called the *integrand*, and a and b are called the *limits of integration*.

Indeed, for monotone functions it is evident that the limit defined above exists from the following fact.

Let f be a monotone function defined on $[a, b]$. The error in approximating the definite integral from a to b by either a left-hand sum or a right-hand sum with n subintervals is bounded by

$$|f(b) - f(a)| \cdot \frac{(b - a)}{n}.$$

Since most of the functions with a bounded domain, that we will investigate in this course, will be either increasing or decreasing over a finite number of intervals we can anticipate that most functions, defined on bounded intervals, that we study will possess an integral.

As a review of some of the concepts from Chapter 3 the following example is given.

Example 6.1.1 Use the Maple V package **student** and the procedure **leftsum** to approximate

$$\int_4^6 \sqrt{1+x^3} dt$$

with an error less than 0.01.

Solution: First analyze the given function over the interval $[4, 6]$. First one needs to find out where the function is monotone. It's a good idea to see a plot of the function. See Figure 5.

```
> f := x -> sqrt(1+x^3);
```

$$f := x \mapsto \sqrt{1+x^3}$$

```
> plot(f(x), x=4..6, y=0..15);
```




Figure 5: Plot of $(1+x^3)^{1/2}$ on $[4, 6]$

From the plot in Figure 5 it appears that the function f is increasing. To be sure check the sign of the derivative. One can either compute the derivative by hand or use Maple V.

```
> diff(f(x), x);
```

$$\frac{3x^2}{2\sqrt{1+x^3}}$$

Note that for x in $[4, 6]$ every factor in the above expression for $f'(x)$ is positive so the derivative is positive and f is increasing. This means that we can estimate the error made in using a left-hand sum by the quantity:

$$|f(6) - f(4)| \cdot \frac{(6-4)}{n}$$

```
> f(6); f(4);
```

$$\frac{\sqrt{217}}{\sqrt{65}}$$

```
> evalf(f(6)-f(4));
```

6.668662112

It may be concluded that the error in estimating the integral

$$\int_4^6 \sqrt{1+x^3} dt$$

by a left-hand sum with n subdivisions is bounded above by

$$(7 \cdot 2)/n.$$

Choose n so that

$$\frac{14}{n} < .01.$$

By taking n larger than $14/.01 = 1400$. The Maple V procedure **leftsum** can be used to check this:

```
> student[leftsum](f(t), t=4..6, 1400);
```

$$\frac{\sum_{i=0}^{1399} \sqrt{1 + \left(4 + \frac{i}{2}\right)^3}}{2}$$

```
> evalf(value("));
```

22.55943031

What numerical value does Maple V give for this integral?

```
> evalf(int(sqrt(1+x^3), x=4..6));
```

22.56419355

Assuming that the last number is equal to the value of the integral to ten digits of accuracy, is the estimate found using **leftsum** valid to the required accuracy?

Example 6.1.2: The above example also solves the following two problems:

1. Estimate the area bounded by the curve

$$y = \sqrt{1+x^3},$$

the x -axis, and the vertical lines $x = 4$ and $x = 6$ to within an accuracy of 0.01.

2. Estimate the average value of the function

$$f(x) = \sqrt{1+x^3},$$

over the interval $[4, 6]$ to within an accuracy 0.01.

Solution: The area under the curve is equal to the integral that was estimated to the desired accuracy in the preceding example. Whereas the average value of the function $f(x)$ over the interval $[4, 6]$ is equal to one half of that integral.

Another result that was discussed in Chapter 3 is the Fundamental Theorem of Calculus:

The Fundamental Theorem of Calculus

If f is defined and continuous on the interval $[a, b]$, and $f(t) = F'(t)$, then

$$\int_a^b f(t) dt = F(b) - F(a).$$

To get an idea of why this is true, fix the the lower end-point of the definite integral and allow the upper end-point to vary. Assume that f is a continuous function defined on $[a, b]$ and $F'(t) = f(t)$ for all $t \in [a, b]$. Define a function H as follows:

```
> H := 'H': f := 'f':
> H := x -> int(f(t), t=a..x);
```

$$H := x \mapsto \int_a^x f(t) dt$$

Since $H(b) = \int_a^b f(t) dt$, and $H(a) = 0$, it is clear that

$$H(b) - H(a) = \int_a^b f(t) dt.$$

It will be shown $H'(t) = f(t)$ on $[a, b]$ which means that the derivative of the two functions H and F agree over an interval. Hence the two functions differ by a constant, say C . But then $H(b) - H(a) = (F(b) + C) - (F(a) + C) = F(b) - F(a)$.

In order to illustrate that Maple V recognizes that $H(b) - H(a)$ is equal to the integral, write:

```
> H(b) - H(a);
```

$$\int_a^b f(t) dt$$

Now take the derivative of H . To do this consider the difference quotient:

$$\frac{H(x+h) - H(x)}{h}.$$

```
> (H(x+h) - H(x)) / h;
```

$$\frac{\int_a^{x+h} f(t) dt - \int_a^x f(t) dt}{h}$$

Rewrite the numerator of this expression as follows:

```
> temp := combine(H(x+h) - H(x));
```

$$temp := \int_x^{x+h} f(t) dt$$

(The validity of the last step will be discussed in the next section.) The function f is continuous on the interval $[a, b]$; so that if

$$a \leq x < x+h \leq b,$$

f is continuous on the interval $[x, x+h]$ also and hence assumes its maximum and minimum values. Let m , and M be the minimum and maximum values of f on this smaller interval, respectively. Let $f(u) = m$ and $f(v) = M$ with u and v points in $[x, x+h]$. Since $m \leq f(t) \leq M$ for $t \in [x, x+h]$, it is reasonable to believe that

```
> L := Int(m, t=x..x+h) <= Int(f(t), t=x..x+h);
```

$$L := \int_x^{x+h} m dt \leq \int_x^{x+h} f(t) dt$$

and

$$\begin{aligned} &> U := \text{Int}(f(t), t=x..x+h) \leq \text{Int}(M, t=x..x+h); \\ U &:= \int_x^{x+h} f(t) dt \leq \int_x^{x+h} M dt \end{aligned}$$

which implies that

$$\begin{aligned} &> a1 := \text{value}(\text{subs}(m=f(u), L/h)); \\ a1 &:= \frac{f(u)(x+h) - f(u)x}{h} \leq \frac{\int_x^{x+h} f(t) dt}{h} \end{aligned}$$

and

$$\begin{aligned} &> a2 := \text{value}(\text{subs}(M=f(v), U/h)); \\ a2 &:= \frac{\int_x^{x+h} f(t) dt}{h} \leq \frac{f(v)(x+h) - f(v)x}{h} \end{aligned}$$

are true. (An illustration of why these hold will be given in the next section.)

Since f is continuous, as h tends to 0, $f(u)$ and $f(v)$ tend to $f(x)$, and the derivative of H at x is bounded by the quantities

$$\begin{aligned} &> \text{limit}(\text{lhs}(a1), h=0) \leq \text{Limit}(\text{rhs}(a1), h=0); \\ f(u) &\leq \lim_{h \rightarrow 0} \frac{\int_x^{x+h} f(t) dt}{h} \end{aligned}$$

and

$$\begin{aligned} &> \text{Limit}(\text{lhs}(a2), h=0) \leq \text{limit}(\text{rhs}(a2), h=0); \\ \lim_{h \rightarrow 0} \frac{\int_x^{x+h} f(t) dt}{h} &\leq f(v) \end{aligned}$$

As h goes to zero $f(u)$ and $f(v)$ become equal to $f(x)$ and $H'(x) = f(x)$. This is what was to be shown.

Remark: The function $F(x)$ in the statement of the Fundamental Theorem is called an *antiderivative* of f . The function $H(x)$ above is also an antiderivative of f . We see from above that two antiderivatives, *i.e.* any two functions with derivative equal to $f(x)$, of the same function differ by a constant.

By restricting to continuous functions $f(x)$ that have an antiderivative $F(x)$ the integral could be defined by

$$\int_a^b f(t) dt = F(b) - F(a),$$

where $F'(x) = f(x)$. The power of the definition of the Riemann Integral is that it does not require the apriori knowledge of an antiderivative and thus does not require the function $f(x)$ to be everywhere continuous. On the other hand, if $f(x)$ is continuous, we can ask if there is a function $F(x)$ such that $F'(x) = f(x)$. We have seen that there always is and it is given to within an additive constant C by the integral formula

$$F(x) = \int_a^x f(t) dt + C.$$

Example 6.1.3 Define and plot an antiderivative of the function f defined by

$$f(x) = \sqrt{3 + x^2}$$

on the interval $[1, 2]$, and the difference quotient at x with $h = 0.2$. Observe that the value of the difference quotient at x and $f(x)$ are close.

Solution: Let $f(x)$ be defined by

```
> f := x -> sqrt(3+x^2);
```

$$f := x \mapsto \sqrt{3 + x^2}$$

An antiderivative for the function is given by the following:

```
> H := x -> int(f(t), t=1..x);
```

$$H := x \mapsto \int_1^x f(t) dt$$

The difference quotient of H is given by

```
> diffquot := x -> int(f(t), t=x..x+h)/h;
```

$$\text{diffquot} := x \mapsto \frac{\int_x^{x+h} f(t) dt}{h}$$

The following Maple V segment plots the function f , the antiderivative, H , and the difference quotient, diffquot (with $h = 0.2$) on the same graph over the interval $[1, 2]$. See Figure 6.

```
> h := 0.2;
```

```
h := .2
```

```
> plot({H, diffquot, f}, 1..2);
```




Figure 6: Plot of $f(t)$, an antiderivative, and a difference quotient

Example 6.1.4 Let

$$f(x) = \sin(x^2)$$

and define

$$H(x) = \int_0^x \sin t^2 dt.$$

1. Verify that $H(x)$ is an antiderivative of $f(x)$ by using Maple V to differentiate $H(x)$.
2. Plot the graph of $H(x)$
3. Find the exact value of $H(0)$.
4. The function $H(x)$ is one of an infinite number of antiderivatives of $f(x)$. Plot the graph of the antiderivative whose value is 1 when $x = 0$. Does any other antiderivative of $f(x)$ have the value 1 at $x = 0$?
5. Plot the graphs of 10 antiderivatives of $f(x)$ simultaneously. Discuss how one antiderivative of $f(x)$ is related to the other.

Solution: Define $f(x)$ by

```
> f := x -> sin(x^2);
```

$$f := x \mapsto \sin(x^2)$$

The antiderivative $H(x)$ is defined by

```
> H := x -> int(f(t), t=0..x);
```

$$H(x) := x \mapsto \int_0^x \sin t^2 dt$$

Before plotting the function consider a few of the other questions in this problem. First of all it, is clear that $H(0) = 0$, since it is equal to the area of a rectangle of zero height and zero width.

```
> H(0);
```

0

Any other antiderivative of $f(x)$ differs from $H(x)$ by a constant. The function $G(x) = H(x) + 1$ is an antiderivative of $f(x)$ since

$$G'(x) = H'(x) = f(x).$$

Furthermore,

$$G(0) = H(0) + 1 = 1.$$

Is there any other antiderivative of $f(x)$ with value 1 at $x = 0$? Suppose that $P(x)$ is an antiderivative of $f(x)$ such that $P(0) = 1$ at $x = 0$. Then since for all x

$$P(x) - H(x) = C$$

for some constant C , we have

$$P(0) - H(0) = 1 - 0 = 1.$$

Therefore,

$$P(x) = H(x) + 1 = G(x),$$



i.e. $P = G$. There is exactly one antiderivative of f that equals 1 at $x = 0$.

Finally, a plot of both $H(x)$ and $G(x)$ on the same graph is given in Figure 7.

```
> G := x -> H(x) + 1;
```

$$G := x \mapsto H(x) + 1$$

```
> plot({H(x), G(x)}, x=-1..1);
```

Figure 7: Plot of two antiderivatives of $\sin(x^2)$.Figure 8: Plot of ten antiderivatives of $\sin(x^2)$.

To complete the solution to this example a Maple V plot of the graphs of the 10 antiderivatives

$$\{H(x) + n/10 \mid n = 0, 1, \dots, 9\}$$

will now be given. Notice that each is parallel to the other in that there is a constant vertical distance between the graphs of any two. See Figure 8

```
> plot( {seq(H(x)+i/10, i=0..9)}, x=-1..1);
```

Exercises 6.1

- Let $f(x) = 5x^3 - 12x^2 + 4x$. Approximate $\int_0^2 f(x)dx$ using Riemann sums corresponding to partitioning $[0, 2]$ into (a) four equal subintervals and evaluating f at the left-hand end point of each subinterval; (b) fifteen equal subintervals and choosing the right-hand endpoint of each subinterval; and (c) use the Fundamental Theorem of Calculus to evaluate the integral exactly.

- (a) Without computing the integral, decide if

$$\int_0^{2\pi} e^{-x} \sin(x) dx$$

is positive or negative; (b) check your answer by numerically evaluating your answer; and (c) explain why you know your answer in (a) is correct and hence step (b) is not necessary.

- Let $f(x) = \arctan x$.

- Using Maple V plot at least 10 antiderivatives of $f(x)$. Explain how the graphs are related.
- Find an antiderivative of $f(x)$ that has value 4 when $x = 2$ and plot its graph. Can there be more than one such antiderivatives? Explain your answer.

- The value of the integral

```
> i1 := Int(cos(sin(x^2)), x=0..4);
```


$$i1 := \int_0^4 \cos(\sin(x^2)) dx$$

is approximately

```
> evalf(i1);
```

3.172054651

How many rectangles are required before the Maple V command **leftsum** returns this answer accurately to three decimal places? (**Hint:** Make a plot of this integrand to see why so many rectangles are needed.)

6.2 General Properties of The Definite Integral

When the definite integral

$$\int_a^b f(t) dt$$

was defined it was assumed that the domain of $f(t)$ was the interval $[a, b]$. It has always been assumed that $a < b$ when discussing the integral. What about the integral when $a \geq b$? For this case, the integral is defined as follows.

Definition

If $a > b$ define

$$\int_a^b f(t) dt = - \int_b^a f(t) dt,$$

and if $a = b$ then

$$\int_a^a f(t) dt = 0.$$

Remark: These definitions would become theorems if the integral had been defined in terms of an antiderivative $F(x)$; since $F(b) - F(a) = -(F(a) - F(b))$, and $F(a) - F(a) = 0$.

Additional Fact About Limits of Integration

For any numbers a , b , and c ,

$$\int_a^c f(t) dt + \int_c^b f(t) dt = \int_a^b f(t) dt.$$

Note that if $f(x)$ is continuous and has an antiderivative $F(x)$ then the above statement follows from

$$\int_a^c f(t) dt + \int_c^b f(t) dt = (F(c) - F(a)) + (F(b) - F(c)) = F(b) - F(a) = \int_a^b f(t) dt.$$

To see why the statement in the box is true for the basic definition assume $a < c < b$ and use the fact that each of the integrals is a limit of sums of the form $\sum f(t_i) \Delta t$ over partitions of the corresponding intervals. One can always choose the partitions of $[a, b]$ in such a way that part of the partition is also a partition of $[a, c]$ and the other part is a partition for $[c, b]$.

The following facts follow similarly.

Facts About Sums and Constant Multiples of the Integrand

Let f and g be integrable functions on $[a, b]$, and let c be a number.

$$\int_a^b (f(t) + g(t))dt = \int_a^b f(t)dt + \int_a^b g(t)dt.$$

$$\int_a^b cf(t)dt = c \int_a^b f(t)dt.$$

If all of the $f(t_i)$ in the sum $\sum f(t_i)\Delta t$ are nonnegative then the sum must also be nonnegative. Hence the following must be true.

Comparing Definite Integrals

If $f(x) \geq 0$, for $x \in [a, b]$, then $\int_a^b f(t)dt \geq 0$.

If $f(x) \leq g(x)$ for $x \in [a, b]$, then

$$\int_a^b f(t)dt \leq \int_a^b g(t)dt.$$

The second statement follows from the first statement and the fact concerning sums of integrals. Since, if $f(x) \leq g(x)$, we have

$$g(x) - f(x) \geq 0$$

and from the first statement we have

$$\int_a^b (g(t) - f(t))dt \geq 0.$$

Upon adding $\int_a^b f(t)dt$ to both sides of the last inequality and applying the fact on the addition of integrals the second statement follows.

Example 6.2.1 Show that

$$0 < \int_0^1 e^{x^2} < 3.$$

Solution: Plot the function e^{x^2} and the constant function $x \rightarrow 3$ on the same graph. See Figure 9.

```
> plot({exp(x^2), 3}, x=0..1);
```

Figure 9 illustrates that for $0 \leq x \leq 1$


$$1 \leq e^{x^2} < 3.$$

More accurately one has:

```
> evalf(exp(0)); evalf(exp(1^2));
```

1.

2.718281828

Figure 9: Plot illustrating $1 \leq \exp(x^2) < 3$

Thus, the increasing function $x \rightarrow e^{x^2}$ satisfies

$$1 \leq e^{x^2} \leq 2.8.$$

By the result on comparing integrals you may conclude that

$$\int_0^1 1 dx \leq \int_0^1 e^{x^2} dx \leq \int_0^1 2.8 dx.$$

Integrating the inequality we obtain

$$1 \leq \int_0^1 e^{x^2} dx \leq 2.8.$$

It follows that, in particular,

$$0 < \int_0^1 e^{x^2} dx < 3.$$

Example 6.2.2 Show that

$$9 < \int_1^4 \sqrt{4+x^2}.$$

Solution: By plotting the graph of the function

```
> f := x -> sqrt(4+x^2);
```


$$x \mapsto \sqrt{4+x^2}$$

One sees that f is increasing (See Figure 10):

```
> plot(f(x), x=1..4);
```

This means that a left-hand sum always produces a lower bound.

```
> student[leftsum](f(t), t=1..4);
```

Figure 10: Plot illustrating $f(t)$ is increasing

$$\frac{3 \sum_{i=0}^3 \sqrt{4 + \left(1 + \frac{3i}{4}\right)^2}}{4}$$

```
> evalf(value("));
```

```
8.933438196
```

The default case when $n = 4$ illustrates that 8.933438196 is a lower bound for the interval, but the problem was to show that 9 is a lower bound. Try $n = 10$.

```
> student[leftsum](f(t), t=1..4, 10);
```

$$\frac{3 \sum_{i=0}^9 \sqrt{4 + \left(1 + \frac{3i}{10}\right)^2}}{10}$$

```
> evalf(");
```

```
9.419027004
```

Thus 9.419027004 is a lower bound for the integral and so is 9.

Observe that the exact result is

```
> Int(f(t), t=1..4): " = value(");
```

$$\int_1^4 \sqrt{4 + t^2} dt = \frac{7\sqrt{5}}{2} + 2 \operatorname{arcsinh}(2) - 2 \operatorname{arcsinh}(1/2)$$

and this is approximately

```
> evalf(rhs("));
```

```
9.751085220
```

Exercises 6.2

1. Suppose that you are given that

$$\int_0^{\pi/2} \cos x^2 dx = .849 \quad \text{and} \quad \int_0^{\pi} \cos x^2 dx = .566.$$

Without using any computers or tables compute:

$$\int_{\pi/2}^{\pi} \cos x^2 dx \quad \text{and} \quad \int_{-\pi/2}^{\pi/2} \cos x^2 dx.$$

2. Using **evalf** and **int** calculate

$$\int_0^3 \exp(-x^3) dx \quad \text{and} \quad \int_0^5 \exp(-x^3) dx.$$

Explain why the two answers appear to be the same even though $\int_3^5 \exp(-x^3) dx > 0$.

3. If f is integrable and $a > 0$, then

$$\int_{-a}^a f(t^2) dt = 2 \int_0^a f(t^2) dt, \quad \int_{-a}^a t f(t^2) dt = 0.$$